Open a note or display dialog with application I work with a colleague editing videos and I wish to trigger a dialog or a note document whenever he opens Final Cut, so I can leave messages for him to do the job. The machine has to stay always on, so I can't use any startup  method. 

It's the same workstation same user account, we share the room in different shifts.

El Capitan 10.11.6 - FinalCut Pro 7 -- He enters after 18pm

Any advice on how to display a message when opening an application?
I don't have “Final Cut.app” so I tested this code with the TextEdit.app.  If you save this code in Script Editor as a stay open application, on first run, this app will display a dialog box for you to input the message that you want the person opening “Final Cut” to read. If the Final Cut closes or unexpectedly quits…. On reopen of Final Cut, the original dialog you previously set, will be displayed again.  When you quit the stay open app and then relaunch it again…. That dialog box will once again appear asking you to enter the message that you want the person opening “Final Cut” to read.

property theApp : "TextEdit" -- Change this
property theMessage : missing value
property appIsRunning : missing value

activate
set theMessage to text returned of (display dialog ¬
    "THE MESSAGE" default answer ¬
    "Message You Want To Display" buttons ¬
    "OK" default button ¬
    "OK" with title ¬
    "Message You Want To Display" with icon 2)

on idle
    repeat
        repeat until appIsRunning is true
            delay 5 -- avoids dialog and theApp (if relaunching) opening at same time
            set appIsRunning to application theApp is running
            if appIsRunning is true then
                activate
                -- Running A Display Dialog Within An Application Tell Block... 
                -- Inherits The Apps Icon To Be Used In The Dialog Window
                tell application theApp to display dialog theMessage buttons {"OK"} default button "OK" with icon 1
            end if
        end repeat
        -- next line necessary to re-trigger the "repeat until" loop again
        set appIsRunning to application theApp is running
        return 5 -- in seconds
    end repeat
end idle